# The LIDL Interaction Description Language

## A language for the description of interaction

Vincent Lecrubier
ONERA DTIM/LAPS
lecrubier@onera.fr

Bruno d'Ausbourg
ONERA DTIM/LAPS
ausbourg@onera.fr

Yamine Aït-Ameur
ENSEEIHT
yamine@enseeiht.fr

## Abstract

This paper describes LIDL, a language dedicated to the specification of interactive systems. LIDL is based on the idea that most programming languages are useful to specify computations, but are not adequate when it comes to specifying interactions. We first introduce the context and the need for new paradigms for interactive systems specification. Then we describe the basic concepts of LIDL, such as Interfaces, Data activation, Interactions, and LIDL program structure an syntax. Some uses of LIDL programs such as verification, code generation and automated testing are then explained. Finally, an interactive system is partially developed using LIDL, as an example use case.

***Categories and Subject Descriptors*** H.5.m. [*Information Interfaces and Presentation (e.g. HCI)*]: Miscellaneous

***Keywords*** Human-Machine Interfaces; Domain-Specific Language; Interactive Systems; Formal Language; Critical Systems

## 1. Introduction

### 1.1 Aeronautical background

A lot of research work have focused on how to design, program and verify functional concerns for critical systems and more particularly aeronautical systems. HMI systems did not benefit from the same attention and efforts.

A significant amount of work has focused on devising models for the development process of software systems in the field of software engineering.

The system development process in critical domains as, for instance, in aeronautics inherited these models. This process is now widely based on the use of standards that take into account the safety and security requirements of the systems under construction. In particular the DO178C standard [3], in aeronautics, defines very strict rules and instructions that must be followed to produce software products, embedded systems and their equipments. The objective is to ensure that the software performs its function with a safety level in accordance with the safety requirements.

Because of the problem stated in Figure 1, the HMI development does not follow the same processes. Nevertheless, in aeronautics, HMI systems are now made up by multiple hardware and software components embedded in aircraft cockpits. These systems are large and complex artifacts that also face tough constraints in terms of usability, security and safety. They support interactive applications that must behave as intended with a high degree of assurance because of their criticity. An error in the software components that implement interactions in these applications may lead to a human or system fault that may have catastrophic effects.

For example, the BEA report [8] about the crash of Rio-Paris AF-447 A330 Airbus establishes that, during the flight, interface system displayed some actions to be performed by the pilot in order to change the pitch of the aircraft and to nose it up while it was stalling. These indications should clearly not have been displayed. Indeed, by following those erroneous displayed instructions the pilot increased the stalling of the aircraft.

In fact, in the industrial context, the development process of critical interactive embedded applications stays very primitive. The usual notations are essentially textual and coding is generally performed from scratch or by reusing previous developments based themselves on textual specifications. In aeronautics, the produced code must be in conformance with the ARINC 661 standard [6]. It may be noticed that some tools recently appeared to enhance the design and coding stages of these systems. But these tools, as for instance Scade Display [27], deal mainly with presentation layers of the systems and do not deal with their complex functional behaviour. In this context, the validation process of the interactive applications is very restricted and poor because it resides practically only in a massive test effort and in expensive evaluation phases at the end of the develop-



*SLE '15*, Month d–d, 20yy, City, ST, Country.

http://dx.doi.org/10.1145/nnnnnnn.nnnnnnn

ment process. Moreover there is no actual formal reference to check the implementation is in conformance with. So new approaches and new paradigms are today needed to help in the development process of critical interactive embedded applications.




**Figure 1.** Critial UIs development is as the intersection of two clashing domains.

### 1.2 Generalization

During the development of LIDL, we realised that the ideas developed in LIDL could also be used successfully to describe other non-critical interactive systems. Indeed, many fields lack appropriate ways to describe interactive systems.

In the field of desktop applications, almost all the popular user-interface toolkits and libraries rely on imperative or object-oriented languages. This means that developers have to use objects in order to describe interactive systems. For instance, Qt or Java programmers are familiar with the notion of listeners, observer, signals, slots... These object-oriented approaches are powerful, but they tend to lead to complex code, sometimes called *spaghetti code* when the complexity becomes intractable.

As expressed in [11], most of the time, developers would like to specify interactions they want their program to perform. But instead, they end up having to code programs that describe how to create and connect groups of objects (widgets, listeners...) that will perform the intended interaction as an emergent behaviour. This adds a lot of complexity and costs, as well as bugs.

The problem is similar in the field of web applications, where many different approaches exists in order to tackle the complexity, but none seems to be satisfactory, as new approaches are created on a regular basis. Hundreds of frameworks try to solve the problem in different ways, but this is a difficult task. As an example, [2] compiles a list of hundreds of different implementations of the same web application.

Finally, the emerging internet of things still misses a language which would allow non-programmers to easily express the interaction of smart objects. Companies such as [1] have devised ways for the general public to describe simple interactions they want their connected objects to perform, but the expressive power of the pseudo-language used is very restrictive.

Overall, it seems that the relative success of general-purpose programming languages in describing interactive systems has eclipsed the urgence of the need for a language specifically designed to describe interactive systems. But as we have seen, the increasing complexity of interactions in a more and more connected digital world is taking general-purpose languages to their limits when it comes to the description of interactive systems.

## 2. The LIDL language

It seems to us that the current state of the art provides no complete solution to the need described in the introduction. The aim of LIDL is to provide a language and tools to deal with this set of problems. LIDL is a relatively simple language based on a small set of features that make sense once put together. The following paragraphs will describe important features that make LIDL a unique language.

### 2.1 Interaction-oriented approach

LIDL programs are defined in a declarative manner, and represent the behaviour of interactive systems whose execution is synchronous, with discrete time steps.

While most programming languages focus on the description of *computations*, the main idea behind LIDL is to describe *interactions*. This is quite a paradigm shift in the sense that many experienced programmers will at first be surprised by the language semantics. However, we argue that LIDL provides an easier way to specify interactive systems, since its main concepts (interfaces and interactions) are more relevant to the field of interactive systems than other programming languages concepts (objects, functions, algorithms...)

A LIDL system represents exactly one interaction. Since interactions can be composed of several simpler interactions connected through interfaces, most non-trivial LIDL programs are a composition of interactions. Actually, LIDL represents everything as interactions. While most programming languages have many different constructions (conditionals, loops, assignments, variables, class declarations, expressions...), LIDL only has two concepts: interactions and interfaces.

***Rationale*** The rationale behind this *interaction-oriented* approach is that LIDL is meant to describe interactive systems. Hence, the only entity that adds value to the description is the interaction. Pushing this interaction-orientation as far as possible led us to the realisation that nothing more than interactions and interfaces is needed, which results in a simple but expressive language.

### 2.2 Synchronous execution

LIDL systems are synchronous. This means that interactions are evaluated at discrete points in time, all at once. As shown in Figure 2, LIDL systems are similar to state machines.

They have a set of input interactions, a set of output interactions, a set of state interactions, and a transition function.

On each step, the transition function uses the previous state and the current input in order to compute the current state and the output of the system.




**Figure 2.** LIDL systems execution

***Rationale*** Synchronous execution tends to simplify the language. The state of the system is explicitly defined for each execution step. This makes reasoning about the code simpler than asynchronous execution. It also maps well to many model checking methods and tools.

### 2.3 Interfaces

An important feature of LIDL is the notion of *interface*. An interface is the combination of two orthogonal aspects: the data type and the data direction. Interfaces are central in LIDL as they have the same role as data types in typed programming languages.

The notion of data type is well known to most programmers. The notion of data direction is also quite easy to understand: the data can either go *in* or go *out*. The notion of interface is hence quite easy to catch, here are a few example of basic interfaces: `Number in`, `Boolean out`, `Text in`...

The same way compound data types exist in other languages, one can express compound interfaces. The syntax to specify compound interfaces is inspired by the Javascript Object Notation (JSON) [17]. Listing 1 shows an example compound interface defined in LIDL.

```
interface Example is
  {
    redSquares      : Square in,
    greenPentagons  : Pentagon in,
    yellowTriangles : Triangle out,
    blueCylinders   : Cylinder out
  }
```

**Listing 1.** LIDL definition of the example interface

Metaphorically, interfaces can be seen as the specification of pipes of specific shapes that allow objects to go in specific directions. Figure 3 shows a way to visualise the example interface of Listing 1.



**Figure 3.** A metaphor of LIDL interfaces as pipes that allow specific data types to flow in specific directions

Every interface has a conjugate interface, which has the same data types, but opposite directions. Two interactive systems can only connect if their interfaces are conjugate. This is the consequence of the natural intuition that the *output* of an entity is the *input* of another one.

***Rationale*** Typed programming languages rely on data types to check the composability of functions and operations. This is convenient when the goal is to describe *computations*. But this is not enough when we try to describe *interactions*. When composing interactions, another very important aspect which is rarely stated explicitly is the *direction* data goes in.

Most programming languages see data as information *stored* at some places. This makes the answer of several important questions difficult to get. When analysing an interactive program, one can (and should!) ask the following questions:

- Where does this piece of data come from?
- What is susceptible of modifying this piece of data?
- Which part of the program have access to this piece of data?
- Does this reference refer to this piece of data?

These questions can become really complex, and are often the cause of many bugs. This is why languages came up with different solutions that allow to answer or simplify these issues: Immutable data types, `const` and `volatile` keywords in the C language, encapsulation in object-oriented programming languages,...

On the other hand, LIDL sees data as information that *flows* from interactions to interactions. The previous questions hence become trivial, because the programmer always knows where the data comes from, and where it is going.

### 2.4 Data activation

Every piece of data in a LIDL program integrates a notion of activation. The implementation is really simple: *all* LIDL

data types are extended with the *inactive* value noted ⊥. For example, the following table shows example values for the basic data types of LIDL:

| Type | Example values |
|---|---|
| `Activation` | ⊥, `T` |
| `Boolean` | ⊥, *true*, *false* |
| `Number` | ⊥, 0, 1, 3.14159 |
| `Text` | ⊥, "*Foo*", "*Bar*", "*Baz*" |

Very simplistically, a flow is represented in LIDL by a piece of data which is almost always active. For example, through an execution, a continuous data flow could have the following trace: {451, 453, 452, 450, 454, ...}. On the other hand an event is represented by a piece of data which is almost always inactive. For example, through an execution, an event could have the following trace: {⊥, ⊥, ⊥, *click*, ⊥, ...}

The notion of activation does not break composability. Here is a compound data type expressed in LIDL : {`x:Number,y:Number`}. This data type is a labelled product data type, similar to a `struct` of the C language. Here are a few example of values of this type: {`x:3,y:2`}, {`x:⊥,y:3`}, ⊥.

***Rationale*** Interactive systems mostly rely on two different paradigms: flow-based representations and event-based representations.

Flow-based representations maps well to systems whose data is defined on *continuous* time intervals, such as physical measurements for example. Examples of flow-based representations include Lustre [20], Scade...

On the other hand, event-based systems maps well to systems whose data is defined on discrete time sets, such as clicks on a button. Examples of event-based representations include most User Interface (UI) Toolkits such as Java Swing, Qt...

Several approaches tried to bridge the gap between flow and event representations [4]. However most approaches are biased toward one paradigm or the other. Interestingly, some approaches treat input and output differently, for example by only allowing discrete inputs (events) and continuous output (status). Figure 4 presents the positioning of different academic approaches regarding this aspect. Shown approaches include [20], [15], [14], [16], [21], [5], [24], [23] and LIDL.

Restriction to a paradigm or the other often prevents natural description of interactive systems, which generally are best described using a mix of both. LIDL proposes a simple way to unify and mix the two paradigms: the notion of data activation.

The notion of data activation is latent in industrial art. Most languages exhibit constructions such as the *null* value, the *maybe* monad, callback functions, listeners, observers, signal slots...

In the context of interactive systems, all these constructions boil down to one unique concept: identify the presence of a piece of data, most of the time a message that has to be received or sent. This is exactly what the data activation feature of LIDL does.




**Figure 4.** Positions of different academic approaches in the flow vs event space.

## 2.5 Interactions

LIDL is a language to describe interactions. The interaction language has a simple syntax, which uses a lot of parentheses. An interaction is a phrase between parentheses, and it composes trivially. Listing 2 shows an example interaction expression, while Figure 5 shows its structure.

```
1 (when (not (powered)) then (turn (light) red))
```

**Listing 2.** An example interaction expression




**Figure 5.** The structure of the example expression. Arrows represent the data flow direction

The semantics of interactions is the most challenging part of LIDL for newcomers, because it is the most disruptive part of the language, since it leverages interfaces and the notion of activation.

Each interaction (i.e. each pair of parentheses, i.e. each block in Figure 5) is attributed a value at each execution step. Depending of the data direction of the interface of the interaction, this value can be defined by the interaction itself (`out`) or by an external interaction (`in`).

Interactions that comply to the `out` interface behave like functions. They output a value, based on their arguments. For example `(not(powered))` receives a boolean `(powered)` and outputs a boolean that is the negation of `(powered)`. This is explained in the following table, which should be easy to understand, with parameters on the left column, and results on the right:

| `(powered)` | `(not(powered))` |
|---|---|
| *true* | *false* |
| *false* | *true* |
| $\bot$ | $\bot$ |

Interactions that comply to the `in` interface behave the opposite way, which is **completely foreign** to programmers. Imagine a function that does not *return* a value based on the arguments it receives, but that *receive* a value, and returns values to its arguments. For example `(turn(light)red)` receives an activation, and outputs a colour `light` which is red when the interaction is active, or $\bot$ the rest of the time. This is summarised in the following table, which will look **unfamiliar** to most programmers, with parameters on the left column, and computation result on the right:

| `(turn(light)red)` | `(light)` |
|---|---|
| $\top$ | $\{red : 255, green : 0, blue : 0\}$ |
| $\bot$ | $\bot$ |

The main advantage of LIDL interaction expressions is that they are very general. Many first-class constructions of other programming languages can be represented as LIDL interactions. As an example, the following table quickly summarises the semantics of the `()=()` interaction. Note that this interaction complies with the `in` interface, indeed, its behaviour consists in sending to the left-hand-side the value it receives on the right-hand-side, *only* when the interaction is active. The control flow, which is implicit in most programming languages, is represented explicitly by the value received by the `()=()` interaction.

| `((x)=(y))` | `(y)` | `(x)` |
|---|---|---|
| $\top$ | 5 | 5 |
| $\top$ | $\bot$ | $\bot$ |
| $\bot$ | 5 | $\bot$ |
| $\bot$ | $\bot$ | $\bot$ |

***Rationale*** First, the resulting syntax is very general, and similar to natural language, with a lot of additional parentheses. These parentheses can be made invisible when editing the code in an appropriate IDE such as those described in [28], which makes reading LIDL as easy as reading natural language.

LIDL has only one construction to construct expressions. While other languages offer many different constructions such as:

- *prefix* operators like `!a`
- *infix* operators like `a + b`
- *postfix* operators like `i++`
- *named* operators like `sin(x)`

Finally, this syntax takes the current evolution of programming languages to its logical conclusion. For example, it generalises the notion of *Named function parameters* as used in Apple's Swift Programming Language, and in Microsoft's C#. Javascript developers also use *argument maps* instead of *positional arguments*. Here is a table comparing different programming languages:

| Language | Expression |
|---|---|
| C, Java | `CalculateBMI(83,185)` |
| Javascript | `CalculateBMI({ weight:83, height:185 })` |
| C#, Swift | `CalculateBMI( weight:83, height:185 )` |
| LIDL (Simple) | `(BMI (83)kg (185)cm)` |
| LIDL (Verbose) | `(Body Mass Index of someone who weights (83) kg and is (185) cm high)` |

Note how LIDL's syntax allows expressions to be much clearer on their semantics. For example, expected units of parameters are only expressed in LIDL.

### 2.6 LIDL programs structure

LIDL programs structure is similar to functional programs structure. Functional programs are represented as a function. A LIDL program is nothing more than an interaction.

The same way that functional programming languages use function signatures to define functions, LIDL use interaction signatures. Since LIDL uses interfaces instead of data types, interaction signatures are described in terms of interfaces.

As an example, here is the signature of the interaction `when()then()` which is instantiated as the root of the example interaction expression of Listing 2:

```
( when (condition: Boolean in)
  then (effect: Activation out)
): Activation in
```

**Listing 3.** The signature of an interaction

The same way that functional programming languages allow to define functions by specifying a signature and the expression it reduces to, LIDL allow to define interactions by specifying a signature and the expression it reduces to.

As an example, here is the definition of the interaction `turn()red` which is used in our example expression of Figure 5:

```
interaction
  (turn (thing: Color out) red): Activation in
is
  ((thing)=({red:(255),green:(0),blue:(0)}))
```

**Listing 4.** Complete LIDL definition of an interaction

Finally, the same way a functional programmer composes functions in order to make more complex functions, a LIDL programmer composes simple interactions in order to make more complex interactions, ending with a final complex interaction: the LIDL program itself.

***Rationale*** LIDL programs structure borrows a lot from functional programming language, because they have a simple structure. LIDL also benefits from features of these languages, such as referential transparency [26], which greatly enhances the readability of programs.

## 3. Use of LIDL programs

LIDL is only a convenient textual way to describe Directed Acyclic Graph (DAG) structures. Indeed, the compiler first expands interactions into base interactions, using definitions. Then it assigns data flow directions using interfaces definitions. This results in a DAG which express the transition function of a state machine. As an example, Figure 6 shows the graph associated with our example expression.

It is really important to notice that the graph shown in Figure 6 is really nothing more than a graph ordering of the graph shown in Figure 5, with data dependency as the ordering relationship. Data dependency is easily inferred from the interfaces.




**Figure 6.** The example expression compiled into a directed acyclic graph

This graph representation is in fact Single Static Assignment (SSA) form [12] of the executable implementing the specified interaction. This form allows different uses such as optimisations, verification, proofs and code generation.

### 3.1 Optimisation

Optimisation can be performed by analysing the graph representation, and generating different execution schemes depending on the requested inputs and outputs, using techniques such as push (data driven evaluation) and pull (demand driven evaluation) as applied to functional reactive programming in [18].

### 3.2 Verification

Verification and proof can be performed by transforming intermediate representation into state machines. The graph representation exactly describes the transition function of such a system, while the state vector is easily derived. It is important to note that the only way for data to persist from one execution step to the next is to be part of a `previous()` interaction. Hence, the state vector is *exactly* the set of interactions which are included in `previous()` interactions. Finally, the generated system has a structure which is very similar to systems generated by other synchronous data flow programming languages such as Scade or Lustre [20]. This potentially allows to leverage the verification tools that have been developed and used for these languages.

### 3.3 Code generation

Code generation has two main objectives: prototype code generation, and production code generation. Both are similar in nature, and are made relatively easy thanks to the intermediate representation. The target languages only have to provide a few features: compound data types, functions, and data types corresponding to LIDL basic data types. At the moment, code generation tools are being developed for two languages: The first is Javascript, in order to enable quick prototyping in a web app, and even some sort of Read-Eval-Print-Loop similar to the one available online for the Elm functional reactive programming language [13]. The other target language is C, as it is probably the most common language in use for critical systems.

### 3.4 Human models and automatic testing

We have seen that LIDL can be used to specify interactions to be performed by computers. It is also a surprisingly convenient way to model interactions to be performed by human agents.

The high abstraction capabilities of LIDL coupled with its close-to-natural-language syntax allows to specify human interactions associated with a system in a very formal way, while remaining similar to a user manual. LIDL descriptions of human interactions are interesting because they bridge the gap between *task models* and *user manuals* [10], being a generalisation of both.

Typically, LIDL developers would code two things: computer-side interactions (e.g a widget behaviour), and their human-side counterparts (e.g. how to use a widget). This approach is similar to test driven development, applied to interactive systems. This formal specification of human interactions enables automatic testing, by executing a system composed of the computer-side interactions on one side, and the human-side interaction on the other, in an approach similar to [7].

Furthermore, LIDL makes it easy to take into account and model human errors and non deterministic behaviour such as those detailed in [9] and [22]. This allows to test

interactive systems even more completely, by simulating the consequences of human errors. Listing 5 shows an example human-side interaction that details how to click on a button, taking into account one error type: omission. The `either()()` interaction represents a non-deterministic choice.

```
1  interaction
2    (click on (theButton: Button)): Activation in
3  is
4    ( either
5      ((theButton.click)=(active))    // Nominal
6      (nothing)                       // Omission !
7    )
```

**Listing 5.** LIDL definition of a potentially faulty human interaction

## 4. Use case

In this section, we will use LIDL to describe parts of a user interface. The user interface in question allows to control the Boiling Water Reactor (BWR) of a nuclear power plant. For the sake of simplicity, we will limit ourselves to an abstract interface as described in [25]. However, LIDL is not restricted to the specification of abstract user interfaces.

LIDL puts an emphasis on reusability. In the use case, this means that we will take advantage of the similarities between components in order to limit the bulk of code. Figure 7 shows common elements in coloured frames, these common elements will be coded as reusable components.


**Figure 7.** A screenshot of the BWR simulator with some common elements outlined in common colors

### 4.1 LIDL implementation of a basic component

To get started, let's look at the implementation of a simple abstract slider, which could be part of a standard abstract widget library for LIDL.

Listing 6 shows the interface that this abstract slider complies to. The abstract slider outputs two things to the user: The value of the slider, concretely implemented by the position of the cursor, and the slider range, concretely implemented by the labels at each end of the slider. The abstract slider has one input from the user: The position that the user wants the slider to be at.

```
1  interface Slider is
2    {
3      value: Number out,
4      range: {min: Number, max: Number} out,
5      selection: Number in
6    }
```

**Listing 6.** The interface of an abstract slider

We could define many interactions that implement this interface. Listing 7 presents one of them. This implementation follows these arbitrary design choices:

- It takes an `enabled` argument that specifies if the slider is enabled or not.
- It takes two arguments to specify the range of the slider.
- In case no value is provided for the slider position, it will initialise as the lower bound of the range.
- It sets the value of the argument `theSelection` when changed by the user, or when the range is changed so that it becomes incompatible with the previous value of the slider.
- It take an argument `constrainedPosition` that allows to programatically set the value of the slider, overriding user input.

Several interactions are used in order to define the slider interaction. For example, note the use of the `()fallbackto()fallbackto...` interaction (lines 16-19). This interaction uses the activation of its arguments, and picks the first argument which is active.

Another important point to notice is the argument named `theSelection` (line 5). Since it is an **out**, it will not be read in order to compute a result. In fact, it will be written to, i.e. a value will be sent to it. This is unlike other arguments that are **in**, which have roles similar to arguments programmers are used to.

By looking at this implementation of the slider, it is really easy to notice that it is a stateful component. Indeed we can see a `previous()` interaction (line 18). The interaction inside the `previous()` is `currentValue`, so `currentValue` is the state variable.

```
1  interaction
2    ( slider (enabled: Activation in)
3      between (min:Number in) and (max:Number in)
4      constrained to (constrainedPosition: Number in)
5      selecting (theSelection: Number out)
6    ): Slider
7  is
8    ((when (enabled)
```

```
    then ({
      value: (currentValue),
      range: ({min: (min),max: (max)}),
      selection: (userInput)
    }))
  behaviour
    ((current value) =
      (((constrainedPosition)
        fallback to (userInput)
        fallback to (previous(currentValue))
        fallback to (min))
      kept between (min) and (max))
    ))
with
  interaction (currentValue):Number ref
  interaction (userInput):Number ref
```

**Listing 7.** The definition of an abstract slider interaction

Listing 8 shows an example use of the slider defined in Listing 7. This instance will always be enabled, because the `enabled` argument is set to the constant `active`. Since the constrained value is set to `inactive`, this instance will allow the user to select a number in the constant range $[0, 2000]$, and the value selected by the user will be sent to a variable named `myValue`.

```
(slider (active) between (0) and (2000)
  constrained to (inactive) selecting (myValue))
```

**Listing 8.** An instance of the abstract slider

### 4.2 LIDL implementation of a compound component

We will describe the components framed in green on Figure 7. These components, that we will call "complex sliders", are composed of:

- A label indicating the purpose of the slider to the user
- A slider allowing the user to select a value. The slider is the one defined in the previous section
- A toggle button to switch between manual and auto modes
- A label indicating the value and units of the selection




**Figure 8.** A screenshot of a complex slider component

Figure 8 shows the concrete implementation of this complex slider, and Listing 9 shows its LIDL interface. Note that it reuses the `Slider` interface defined in the previous section, as well as other interfaces.

```
interface ComplexSlider is
  {
    title: Label,
    slider: Slider,
    toggle: ToggleButton,
    value: Label
  }
```

**Listing 9.** The interface of the complex slider

Listing 10 shows an implementation of this complex slider.

```
interaction (
  complex slider
  named (title: Text in)
  between (min:Number in) and (max:Number in)
  (units:Text in)
  constrained to (constrainedPosition: Number in)
  selecting (theSelection: Number out)
  requesting (mode: Activation out) automation
  ):ComplexSlider
is
  ({
    title: (
      label (active)
      displaying (title)),
    slider: (
      slider (active)
      between (min) and (max)
      constrained to (constrainedPosition)
      selecting (theSelection)),
    toggle: (
      toggle (active)
      pushed (when(constrainedPosition))
      displaying ("A")
      toggling (mode)),
    value: (
      label (active)
      displaying ((theSelection) " " (units)) )
  })
```

**Listing 10.** Definition of a complex slider interaction

Listing 11 shows an example instance of this complex slider, corresponding to the concrete implementation depicted in Figure 8.

```
( complex slider
  named ("Control Rods Level")
  between (0) and (100) ("%")
  constrained to (controlRodsAutoValue)
  selecting (controlRodsLevel)
  requesting (controlRodsAutoMode) automation
)
```

**Listing 11.** An instance of the complex slider interaction

## 5. Conclusion

This paper presented a quick overview of LIDL, a language dedicated to the description of interactions, and a use case. The use case showed that LIDL allows to specify safe complex behaviour. In particular, it is noteworthy that, as compared to other approaches, the LIDL way of thinking as two consequences:

- Removing duplicate or boilerplate code as seen in other languages, such as getter/setters and observer pattern functions. This is noticeable by the relatively small size of LIDL programs.

- Forcing designers into thinking about the actual interaction, enforcing to explicitly define aspects that are usually implicit or merged into objects whose semantics are not clear. This is noticeable in the slider example (Listing 7), where an explicit distinction is made between the user input and the slider current value. This explicit distinction allows to have a sane behaviour, even when the slider range is dynamically changed, while keeping the code simple.

LIDL is only a language. Architectural concepts that fit with LIDL are being developed, but not detailled in this paper. The architectural ideas behind LIDL converge with those recently presented in [19] and similar approaches around uni-directional data flow. A general framework for the specification of abstraction levels of interactive systems inspired by [29] is being developed in parallel with LIDL.

## References


[1] Ifttt. URL www.ifttt.com.

[2] Todomvc.com. URL www.todomvc.com.

[3] DO178C. Software Consideration in Airborn Systems and Equipment Certification, release c, 2012. RTCA,Inc.

[4] G. D. Abowd and A. J. Dix. Integrating status and event phenomena in formal specifications of interactive systems. *Proceedings of the ACM SIGSOFT Symposium on Foundations of Software Engineering*, 22:23, December 1994.

[5] G. A. Agha. Actors: a model of concurrent computations in distributed systems. *MIT Press*, 1986.

[6] *Specification 661*. ARINC, supplement 5, draft 1 edition, March 2012. URL http://www.aviation-ia.com/aeec/projects/cds/index.html.

[7] E. Barboni, J.-F. Ladry, D. Navarre, P. Palanque, and M. Winckler. Beyond modelling: An integrated environment supporting co-execution of tasks and systems models. In *Proceedings of the 2Nd ACM SIGCHI Symposium on Engineering Interactive Computing Systems*, EICS '10, pages 165–174, New York, NY, USA, 2010. ACM. ISBN 978-1-4503-0083-4. . URL http://doi.acm.org/10.1145/1822018.1822043.

[8] BEA. Rapport final sur l'accident survenu le 1er juin 2009 à l'Airbus A330-203 immatriculé F-GZCP exploité par Air France, vol AF 447 Rio de Janeiro - Paris. Technical report, Direction Générale de l'Aviation Civile, Juillet 2012. http://www.bea.aero/docspa/2009/f-cp090601/pdf/f-cp090601.pdf.

[9] M. Bolton and E. Bass. Evaluating human-automation interaction using task analytic behavior models, strategic knowledge-based erroneous human behavior generation, and model checking. In *Systems, Man, and Cybernetics (SMC), 2011 IEEE International Conference on*, pages 1788 –1794, oct. 2011. .

[10] J. Bowen and S. Reeves. Modelling user manuals of modal medical devices and learning from the experience. In *Proceedings of the 4th ACM SIGCHI Symposium on Engineering Interactive Computing Systems*, EICS '12, pages 121–130, New York, NY, USA, 2012. ACM. ISBN 978-1-4503-1168-7. . URL http://doi.acm.org/10.1145/2305484.2305505.

[11] S. Chatty. Programs = data + algorithms + architecture, and consequences for interactive software. In S. Verlag, editor, *Proceedings of the IFIP conference on Engineering Interactive Systems (EIS 2007)*, volume Lecture Notes in Computer Science volume 4940, 2007. URL http://lii-enac.fr/articles/chatty-eis-2007.pdf.

[12] R. Cytron, J. Ferrante, B. K. Rosen, M. N. Wegman, and F. K. Zadeck. Efficiently computing static single assignment form and the control dependence graph. *ACM Transactions on Programming Languages and Systems*, 13(4):451–490, Oct 1991. URL http://doi.acm.org/10.1145/115372.115320.

[13] E. Czaplicki. Elm: Concurrent frp for functional guis. 2012.

[14] A. Dix and C. Runciman. Abstract models of interactive systems. In *Proceedings of the HCI'85 Conference on People and Computers: Designing the Interface*, pages 13–22, 1985.

[15] A. J. Dix. *Formal methods for interactive systems*. Academic Press, 1991.

[16] D. Duke and M. Harrison. Abstract interaction objects. *Computer Graphics Forum*, 12(3):25–36, 1993. ISSN 1467-8659. . URL http://dx.doi.org/10.1111/1467-8659.1230025.

[17] ECMA. The json data interchange format. October 2013. URL http://www.ecma-international.org/publications/files/ECMA-ST/ECMA-404.pdf.

[18] C. M. Elliott. Push-pull functional reactive programming. In *Proceedings of the 2Nd ACM SIGPLAN Symposium on Haskell*, Haskell '09, pages 25–36, New York, NY, USA, 2009. ACM. ISBN 978-1-60558-508-6. . URL http://doi.acm.org/10.1145/1596638.1596643.

[19] Facebook. React - a javascript library for building user interfaces, 2013. URL http://facebook.github.io/react/.

[20] N. Halbwachs, P. Caspi, P. Raymond, and D. Pilaud. The synchronous dataflow programming language Lustre. In *Proceedings of IEEE*, number 9 in 79, pages 1305–1320, September 1991.

[21] D. Harel. Statecharts: A visual formalism for complex systems. *Science of Computer Programming*, 8(3):231–274, 1987.

[22] C. Martinie and P. Palanque. Fine grain modeling of task deviations for assessing qualitatively the impact of both system failures and human error on operator performance. In *2014 AAAI Spring Symposium Series*, 2014.

[23] D. Navarre, P. Palanque, J.-F. Ladry, and E. Barboni. Icos: A model-based user interface description technique dedicated to interactive systems addressing usability, reliability and scalability. *ACM Trans. Comput.-Hum. Interact.*, 16:18:1–18:56, November 2009. ISSN 1073-0516. . URL http://doi.acm.org/10.1145/1614390.1614393.

[24] F. Paternò and G. Faconti. On the use of LOTOS to describe graphical interaction. In *Proceedings of the HCI'92 Conference on People and Computers*, pages 155–173, 1992.

[25] F. Paterno, C. Santoro, and L. D. Spano. Maria: A universal, declarative, multiple abstraction-level language for service-oriented applications in ubiquitous environments. *ACM Trans. Comput.-Hum. Interact.*, 16(4):19:1–19:30, Nov. 2009. ISSN 1073-0516. . URL http://doi.acm.org/10.1145/1614390.1614394.

[26] H. Søndergaard and P. Sestoft. Referential transparency, definiteness and unfoldability. *Acta Informatica*, 27(6):505–517, 1990. ISSN 0001-5903. . URL http://dx.doi.org/10.1007/BF00277387.

[27] E. Technologies. Scade display, 2011. URL http://www.esterel-technologies.com/products/scade-display.

[28] M. Voelter, J. Siegmund, T. Berger, and B. Kolb. Towards user-friendly projectional editors. In B. Combemale, D. Pearce, O. Barais, and J. Vinju, editors, *Software Language Engineering*, volume 8706 of *Lecture Notes in Computer Science*, pages 41–61. Springer International Publishing, 2014. ISBN 978-3-319-11244-2. . URL http://dx.doi.org/10.1007/978-3-319-11245-9_3.

[29] P. Zave and J. Rexford. The geomorphic view of networking: A network model and its uses. In *Proceedings of the 7th Workshop on Middleware for Next Generation Internet Computing*, MW4NG '12, pages 1:1–1:6, New York, NY, USA, 2012. ACM. ISBN 978-1-4503-1607-1. . URL http://doi.acm.org/10.1145/2405178.2405179.