

The LIDL Interaction Description Language

Vincent Lecrubier
ONERA DTIM/LAPS
Toulouse, France
lecrubier@onera.fr

Bruno d'Ausbourg
ONERA DTIM/LAPS
Toulouse, France
ausbourg@onera.fr

Yamine Aït-Ameur
ENSEEIH
Toulouse, France
yamine@enseeiht.fr

ABSTRACT

This paper describes LIDL, a language dedicated to the specification of interactive systems. LIDL is based on the idea that most programming languages are useful to specify computations, but are not adequate when it comes to specifying interactions. We first introduce the context and the need for new paradigms for interactive systems specification. Then we describe the basic concepts of LIDL, such as Interfaces, Data activation, Interactions, and LIDL program structure. Some uses of LIDL programs such as verification and code generation are then explained. Finally, a boiling water nuclear reactor user interface is partially developed using LIDL, as an example use case.

ACM Classification Keywords

H.5.m. Information Interfaces and Presentation (e.g. HCI): Miscellaneous

Author Keywords

Human-Machine Interfaces; Domain-Specific Language; Interactive Systems; Formal Language; Critical Systems

INTRODUCTION

A lot of research work have focused on how to design, program and verify functional concerns for critical systems and more particularly aeronautical systems. HMI systems did not benefit from the same attention and efforts.

A significant amount of work has focused on devising models for the development process of software systems in the field of software engineering.

The system development process in critical domains as, for instance, in aeronautics inherited these models. This process is now widely based on the use of standards that take into account the safety and security requirements of the systems under construction. In particular the DO178C standard [1], in aeronautics, defines very strict rules and instructions that must be followed to produce software products, embedded systems and their equipments. The objective is to ensure that the software performs its function with a safety level in accordance with the safety requirements.

Paste the appropriate copyright statement here. ACM now supports three different copyright statements:

- ACM copyright: ACM holds the copyright on the work. This is the historical approach.
- License: The author(s) retain copyright, but ACM receives an exclusive publication license.
- Open Access: The author(s) wish to pay for the work to be open access. The additional fee must be paid to ACM.

This text field is large enough to hold the appropriate release statement assuming it is single spaced.

Every submission will be assigned their own unique DOI string to be included here.

Because of the problem stated in Figure 1, the HMI development does not follow the same processes. Nevertheless, in aeronautics, HMI systems are now made up by multiple hardware and software components embedded in aircraft cockpits. These systems are large and complex artifacts that also face tough constraints in terms of usability, security and safety. They support interactive applications that must behave as intended with a high degree of assurance because of their criticality. An error in the software components that implement interactions in these applications may lead to a human or system fault that may have catastrophic effects.

For example, the BEA report [5] about the crash of Rio-Paris AF-447 A330 Airbus establishes that, during the flight, interface system displayed some actions to be performed by the pilot in order to change the pitch of the aircraft and to nose it up while it was stalling. These indications should clearly not have been displayed. Indeed, by following those erroneous displayed instructions the pilot increased the stalling of the aircraft.

In fact, in the industrial context, the development process of critical interactive embedded applications stays very primitive. The usual notations are essentially textual and coding is generally performed from scratch or by reusing previous developments based themselves on textual specifications. In aeronautics, the produced code must be in conformance with the ARINC 661 standard [4]. It may be noticed that some tools recently appeared to enhance the design and coding stages of these systems. But these tools, as for instance Scade Display [18], deal mainly with presentation layers of the systems and do not deal with their complex functional behaviour. In this context, the validation process of the interactive applications is very restricted and poor because it resides practically only in a massive test effort and in expensive evaluation phases at the end of the development process. Moreover there is no actual formal reference to check the implementation is in conformance with. So new approaches and new paradigms are today needed to help in the development process of critical interactive embedded applications.

THE LIDL LANGUAGE

It seems to us that the current state of the art provides no complete solution to the need described in the previous section. The aim of LIDL is to provide a language and tools to deal with this set of problems.

Informal presentation

While most programming languages focus on the description of *computations*, the main idea behind LIDL is to describe




Figure 1. Critical UIs development is as the intersection of two clashing domains.

interactions. This is quite a paradigm shift in the sense that many experienced programmers will at first be surprised by the language semantics. However, we argue that LIDL provides an easier way to specify interactive systems, since its main concepts (interfaces and interactions) are more relevant to the field of interactive systems than other programming languages concepts (objects, functions, algorithms...)

LIDL programs are defined in a declarative manner, and represent interactive systems whose execution is synchronous.

Interfaces

Typed programming languages rely on data types to check the composability of functions and operations. This is convenient when the goal is to describe *computations*. But this is not enough when we try to describe *interactions*. When composing interactions, another very important aspect which is rarely stated is the *direction* data goes in.

As an answer to that matter, an important feature of LIDL is the notion of *interface*. An interface is the combination of two orthogonal aspects: the data type and the data direction.

The notion of data type is well known to most programmers. The notion of data direction is also quite easy to understand: the data can either go *in* or go *out*. The notion of interface is hence quite easy to catch, here are a few example of basic interfaces: Number *in*, Boolean *out*, Text *in*...

The same way compound data types exist, one can express compound interfaces. The syntax to specify compound interfaces is inspired by the Javascript Object Notation (JSON) [10]. Listing 1 shows an example compound interface defined in LIDL.

```
1 interface Example is
2 {
3   redSquares      : Square in,
4   greenPentagons  : Pentagon in,
5   yellowTriangles : Triangle out,
6   blueCylinders   : Cylinder out
7 }
```

Listing 1. LIDL definition of the example interface

Metaphorically, interfaces can be seen as the specification of pipes of specific shapes that allow objects to go in specific

directions. Figure 2 shows a way to visualise the example interface of Listing 1.




Figure 2. A metaphor of interfaces as pipes that allow specific data types to flow in specific directions

Every interface has a conjugate interface, which has the same data types, but opposite directions. Two interactive systems can only connect if their interfaces are conjugate. This is the consequence of the natural intuition that the output of an entity is the input of another one.

Interfaces are central in LIDL as they have the same role as data types in typed programming languages.

Data activation

Interactive systems rely on two different paradigms: flow-based representations and event-based representations.

Flow-based representations maps well to systems whose data is defined on *continuous* time intervals, such as the pressure inside a reactor. Examples of flow-based representations include Lustre [13], Scade...

On the other hand, event-based systems maps well to systems whose data is defined on discrete time sets, such as clicks on a button. Examples of event-based representations include most User Interface (UI) Toolkits such as Java Swing, Qt...

Several approaches tried to bridge the gap between flow and event representations [2]. However most approaches are biased toward one paradigm or the other. Interestingly, some approaches treat input and output differently, for example by only allowing discrete inputs (events) and continuous output (status). Figure 3 presents the positioning of different academic approaches regarding this aspect. Shown approaches include [13], [8], [7], [9], [14], [3], [16], [15] and LIDL.

Restriction to a paradigm or the other often prevents natural description of interactive systems, which generally are best described using a mix of both. LIDL proposes a simple way to unify and mix the two paradigms: the notion of data activation.

The notion of data activation is latent in industrial art. Most languages exhibit constructions such as the *null* value, the *maybe* monad, callback functions, listeners, observers, signal slots...




Figure 3. Positions of different academic approaches in the flow vs event space.

In the context of interactive systems, all these constructions boil down to one unique concept: identify the presence of a piece of data, most of the time a message that has to be received or sent. This is exactly what the data activation feature of LIDL does.

Without exception, every piece of data in a LIDL program integrates a notion of activation. The implementation is really simple: *all* LIDL data types are extended with the *inactive* value noted \perp . For example, the following table shows example values for the basic data types of LIDL:

Type	Example values
Activation	\perp , \top
Boolean	\perp , <i>true</i> , <i>false</i>
Number	\perp , 0, 1, 3.14159
Text	\perp , "Foo", "Bar", "Baz"

Very simplistically, a flow is represented in LIDL by a piece of data which is almost always active. For example, through an execution, the pressure in a reactor would have the following trace: {451, 453, 452, 450, 454, ...}. On the other hand an event is represented by a piece of data which is almost always inactive. For example, through an execution, clicks on a button would have the following trace: { \perp , \perp , \perp , *click*, \perp , ...}

The notion of activation does not break composability. Here is a compound data type expressed in LIDL : {*x*:Number, *y*:Number}. This data type is a labelled product data type, similar to a `struct` of the C language. Here are a few example of values of this type: {*x*:3, *y*:2}, {*x*: \perp , *y*:3}, \perp .

Interactions

LIDL is a language to describe interactions. The interaction language has a simple syntax, which uses a lot of parentheses. An interaction is a phrase between parentheses, and it

composes trivially. Listing 2 shows an example interaction expression, while Figure 4 shows its structure.

```
1 (when(not (powered)) then(turn(light) red) )
```

Listing 2. An example interaction expression




Figure 4. The structure of the example expression. Arrows represent the data flow direction

The semantics of interactions is the most challenging part of LIDL for newcomers, because it is the most disruptive part of the language, since it leverages interfaces and the notion of activation.

Each interaction (i.e. each pair of parentheses, i.e. each block in Figure 4) is attributed a value at each execution step. Depending of the data direction of the interface of the interaction, this value can be defined by the interaction itself (**out**) or by an external interaction (**in**).

Interactions that comply to the **out** interface behave like functions. They output a value, based on their arguments. For example (not (powered)) receives a boolean (powered) and outputs a boolean that is the negation of (powered). This is explained in the following table, which should be easy to understand, with parameters on the left column, and results on the right:

(powered)	(not (powered))
<i>true</i>	<i>false</i>
<i>false</i>	<i>true</i>
\perp	\perp

Interactions that comply to the **in** interface behave the opposite way, which is **completely foreign** to programmers. Imagine a function that does not *return* a value based on the arguments it receives, but that *receive* a value, and returns values to its arguments. For example (turn(light) red) receives an activation, and outputs a colour *light* which is red when the interaction is active, or \perp the rest of the time. This is summarised in the following table, which will look **unfamiliar** to most programmers, with parameters on the left column, and computation result on the right:

(turn(light) red)	(light)
\top	{ <i>red</i> : 255, <i>green</i> : 0, <i>blue</i> : 0}
\perp	\perp

LIDL programs structure

LIDL programs structure is similar to functional programs structure. Functional programs are represented as a function. A LIDL program is nothing more than an interaction.

The same way that functional programming languages use function signatures to define functions, LIDL use interaction signatures. Since LIDL uses interfaces instead of data types, interaction signatures are described in terms of interfaces.

As an example, here is the signature of the interaction `when()` `then()` which is instantiated as the root of the example interaction expression of Listing 2:

```
1 ( when (condition: Boolean in)
2   then (effect: Activation out)
3 ): Activation in
```

Listing 3. The signature of an interaction

The same way that functional programming languages allow to define functions by specifying a signature and the expression it reduces to, LIDL allow to define interactions by specifying a signature and the expression it reduces to.

As an example, here is the definition of the interaction `turn()` `red` which is used in our example expression of Figure 4:

```
1 interaction
2 (turn (thing: Color out) red): Activation in
3 is
4 ((thing)=({red: (255),green: (0),blue: (0)}))
```

Listing 4. Complete LIDL definition of an interaction

Finally, the same way a functional programmer composes functions in order to make more complex functions, a LIDL programmer composes simple interactions in order to make more complex interactions, ending with a final complex interaction: the LIDL program itself.

Use of LIDL programs

LIDL is only a convenient textual way to describe Directed Acyclic Graph (DAG) structures. Indeed, the compiler first expands interactions into base interactions, using definitions. Then it assigns data flow directions using interfaces definitions. This results in a DAG which express the transition function of a state machine. As an example, Figure 5 shows the graph associated with our example expression.

It is really important to notice that the graph shown in Figure 5 is really nothing more than a graph ordering of the graph shown in Figure 4, with data dependency as the ordering relationship. Data dependency is easily inferred from the interfaces.

This graph representation is in fact Single Static Assignment (SSA) form [6] of the executable implementing the specified interaction. This form allows different uses such as optimisations, verification, proofs and code generation.

Optimisation can be performed by analysing the graph representation, and generating different execution schemes depending on the requested inputs and outputs, using techniques such as push (data driven evaluation) and pull (demand driven evaluation) as applied to functional reactive programming in [11].




Figure 5. The example expression compiled into a directed acyclic graph

Verification and proof can be performed by transforming intermediate representation into state machines. The graph representation exactly describes the transition function of such a system, while the state vector is easily derived. It is important to note that the only way for data to persist from one execution step to the next is to be part of a `previous()` interaction. Hence, the state vector is *exactly* the set of interactions which are included in `previous()` interactions.

Code generation has two main objectives: prototype code generation, and production code generation. Both are similar in nature, and are made relatively easy thanks to the intermediate representation. The target languages only have to provide a few features: compound data types, functions, and data types corresponding to LIDL basic data types.

USE CASE

In this section, we will use LIDL to describe the Boiling Water Reactor (BWR) use case. For the sake of simplicity, we will limit ourselves to an abstract interface as described in [17]. However, LIDL is not restricted to the specification of abstract user interfaces.

LIDL puts an emphasis on reusability. In the use case, this means that we will take advantage of the similarities between components in order to limit the bulk of code. Figure 6 shows common elements in coloured frames, these common elements will be coded as reusable components.

LIDL implementation of a basic component

To get started, let's look at the implementation of a simple abstract slider, which could be part of a standard abstract widget library for LIDL.

Listing 5 shows the interface that this abstract slider complies to. The abstract slider outputs two things to the user: The value of the slider, concretely implemented by the position of the cursor, and the slider range, concretely implemented by the labels at each end of the slider. The abstract slider has one input from the user: The position that the user wants the slider to be at.

```
1 interface Slider is
2 {
3   value: Number out,
```




Figure 6. A screenshot of the BWR simulator with some common elements outlined in common colors

```

4   range: {min: Number, max: Number} out,
5   selection: Number in
6 }

```

Listing 5. The interface of an abstract slider

We could define many interactions that implement this interface. Listing 6 presents one of them. This implementation follows these arbitrary design choices:

- It takes an `enabled` argument that specifies if the slider is enabled or not.
- It takes two arguments to specify the range of the slider.
- In case no value is provided for the slider position, it will initialise as the lower bound of the range.
- It sets the value of the argument `theSelection` when changed by the user, or when the range is changed so that it becomes incompatible with the previous value of the slider.
- It takes an argument `constrainedPosition` that allows to programmatically set the value of the slider, overriding user input.

Several interactions are used in order to define the slider interaction. For example, note the use of the `() fallbackto()` `fallbackto...` interaction (lines 16-19). This interaction uses the activation of its arguments, and picks the first argument which is active.

Another important point to notice is the argument named `theSelection` (line 5). Since it is an `out`, it will not be read in order to compute a result. In fact, it will be written to, i.e. a value will be sent to it. This is unlike other arguments that are `in`, which have roles similar to arguments programmers are used to.

By looking at this implementation of the slider, it is really easy to notice that it is a stateful component. Indeed we can see a `previous()` interaction (line 18). The interaction inside the `previous()` is `currentValue`, so `currentValue` is the state variable.

```

1 interaction
2 ( slider (enabled: Activation in)
3   between (min: Number in) and (max: Number in)
4   constrained to (constrainedPosition: Number in)
5   selecting (theSelection: Number out)
6 ): Slider
7 is
8 ( (when (enabled)
9   then ({
10    value: (currentValue),
11    range: ({min: (min), max: (max)}),
12    selection: (userInput)
13  }))
14 behaviour
15 ((current value)=
16  ((constrainedPosition)
17   fallback to (userInput)
18   fallback to (previous(currentValue))
19   fallback to (min))
20   kept between (min) and (max))
21 ))
22 with
23 interaction (currentValue): Number ref
24 interaction (userInput): Number ref

```

Listing 6. The definition of an abstract slider interaction

Listing 7 shows an example use of the slider defined in Listing 6. This instance will always be enabled, because the `enabled` argument is set to the constant `active`. Since the constrained value is set to `inactive`, this instance will allow the user to select a number in the constant range `[0, 2000]`, and the value selected by the user will be sent to a variable named `myValue`.

```

1 (slider (active) between (0) and (2000)
2   constrained to (inactive) selecting (myValue))

```

Listing 7. An instance of the abstract slider

LIDL implementation of a compound component

We will describe the components framed in green on Figure 6. These components, that we will call “complex sliders”, are composed of:

- A **label** indicating the purpose of the slider to the user
- A **slider** allowing the user to select a value. The slider is the one defined in the previous section
- A **toggle button** to switch between manual and auto modes
- A **label** indicating the value and units of the selection




Figure 7. A screenshot of a complex slider component

Figure 7 shows the concrete implementation of this complex slider, and Listing 8 shows its LIDL interface. Note that it

reuses the `Slider` interface defined in the previous section, as well as other interfaces.

```
1 interface ComplexSlider is
2 {
3     title: Label,
4     slider: Slider,
5     toggle: ToggleButton,
6     value: Label
7 }
```

Listing 8. The interface of the complex slider

Listing 9 shows an implementation of this complex slider.

```
1 interaction (
2     complex slider
3     named (title: Text in)
4     between (min: Number in) and (max: Number in)
5     (units: Text in)
6     constrained to (constrainedPosition: Number in)
7     selecting (theSelection: Number out)
8     requesting (mode: Activation out) automation
9 ) : ComplexSlider
10 is
11 ({
12     title: (
13         label (active)
14         displaying (title)),
15     slider: (
16         slider (active)
17         between (min) and (max)
18         constrained to (constrainedPosition)
19         selecting (theSelection)),
20     toggle: (
21         toggle (active)
22         pushed (when (constrainedPosition))
23         displaying ("A")
24         toggling (mode)),
25     value: (
26         label (active)
27         displaying ((theSelection) " " (units)) )
28 })
```

Listing 9. Definition of a complex slider interaction

Listing 10 shows an example instance of this complex slider, corresponding to the concrete implementation depicted in Figure 7.

```
1 ( complex slider
2     named ("Control Rods Level")
3     between (0) and (100) ("%")
4     constrained to (controlRodsAutoValue)
5     selecting (controlRodsLevel)
6     requesting (controlRodsAutoMode) automation
7 )
```

Listing 10. An instance of the complex slider interaction

CONCLUSION

This paper presented a quick overview of LIDL, a language dedicated to the description of interactions, and a use case. The use case showed that LIDL allows to specify safe complex behaviour. In particular, it is noteworthy that, as compared to other approaches, the LIDL way of thinking as two consequences:

- Removing duplicate or boilerplate code as seen in other languages, such as getter/setters and observer pattern functions. This is noticeable by the relatively small size of LIDL programs.

- Forcing designers into thinking about the actual interaction, enforcing to explicitly define aspects that are usually implicit or merged into objects whose semantics are not clear. This is noticeable in the slider example (Listing 6), where an explicit distinction is made between the user input and the slider current value. This explicit distinction allows to have a sane behaviour, even when the slider range is dynamically changed, while keeping the code simple.

LIDL is only a language. Architectural concepts that fit with LIDL are being developed, but not detailed in this paper. The architectural ideas behind LIDL converge with those recently presented in [12] and similar approaches around unidirectional data flow. A general framework for the specification of abstraction levels of interactive systems inspired by [19] is being developed in parallel with LIDL.

REFERENCES

1. 2012. DO178C. Software Consideration in Airborne Systems and Equipment Certification, release C. (2012). RTCA, Inc.
2. Gregory D Abowd and Alan J Dix. 1994. Integrating status and event phenomena in formal specifications of interactive systems. *Proceedings of the ACM SIGSOFT Symposium on Foundations of Software Engineering 22* (December 1994), 23.
3. G. A. Agha. 1986. Actors: a model of concurrent computations in distributed systems. *MIT Press* (1986).
4. ARINC 2012. *Specification 661* (supplement 5, draft 1 ed.). ARINC. <http://www.aviation-ia.com/aeec/projects/cds/index.html>
5. BEA. 2012. *Rapport final sur l'accident survenu le 1er juin 2009 à l'Airbus A330-203 immatriculé F-GZCP exploité par Air France, vol AF 447 Rio de Janeiro - Paris*. Technical Report. Direction Générale de l'Aviation Civile. <http://www.bea.aero/docs/2009/f-cp090601/pdf/f-cp090601.pdf>.
6. Ron Cytron, Jeanne Ferrante, Barry K. Rosen, Mark N. Wegman, and F. Kenneth Zadeck. 1991. Efficiently Computing Static Single Assignment Form and the Control Dependence Graph. *ACM Transactions on Programming Languages and Systems* 13, 4 (Oct 1991), 451–490. <http://doi.acm.org/10.1145/115372.115320>
7. A. Dix and C. Runciman. 1985. Abstract models of interactive systems. In *Proceedings of the HCI'85 Conference on People and Computers: Designing the Interface*. 13–22.
8. Alan John Dix. 1991. *Formal methods for interactive systems*. Academic Press.
9. D.J. Duke and M.D. Harrison. 1993. Abstract Interaction Objects. *Computer Graphics Forum* 12, 3 (1993), 25–36. DOI: <http://dx.doi.org/10.1111/1467-8659.1230025>

10. ECMA. 2013. The JSON Data Interchange Format. (October 2013). <http://www.ecma-international.org/publications/files/ECMA-ST/ECMA-404.pdf>
11. Conal M. Elliott. 2009. Push-pull Functional Reactive Programming. In *Proceedings of the 2Nd ACM SIGPLAN Symposium on Haskell (Haskell '09)*. ACM, New York, NY, USA, 25–36. DOI : <http://dx.doi.org/10.1145/1596638.1596643>
12. Facebook. 2013. React - a JavaScript library for building user interfaces. (2013). <http://facebook.github.io/react/>
13. Nicolas Halbwachs, Paul Caspi, Pascal Raymond, and Daniel Pilaud. 1991. The synchronous dataflow programming language Lustre. In *Proceedings of IEEE (79)*. 1305–1320.
14. D. Harel. 1987. Statecharts: A visual formalism for complex systems. *Science of Computer Programming* 8, 3 (1987), 231–274.
15. David Navarre, Philippe Palanque, Jean-Francois Ladry, and Eric Barboni. 2009. ICOs: A model-based user interface description technique dedicated to interactive systems addressing usability, reliability and scalability. *ACM Trans. Comput.-Hum. Interact.* 16, Article 18 (November 2009), 56 pages. Issue 4. DOI : <http://dx.doi.org/10.1145/1614390.1614393>
16. F. Paternò and G. Faconti. 1992. On the use of LOTOS to describe graphical interaction. In *Proceedings of the HCI'92 Conference on People and Computers*. 155–173.
17. Fabio Paterno, Carmen Santoro, and Lucio Davide Spano. 2009. MARIA: A Universal, Declarative, Multiple Abstraction-level Language for Service-oriented Applications in Ubiquitous Environments. *ACM Trans. Comput.-Hum. Interact.* 16, 4, Article 19 (Nov. 2009), 30 pages. DOI : <http://dx.doi.org/10.1145/1614390.1614394>
18. Esterel Technologies. 2011. Scade Display. (2011). <http://www.esterel-technologies.com/products/scade-display>
19. Pamela Zave and Jennifer Rexford. 2012. The Geomorphic View of Networking: A Network Model and Its Uses. In *Proceedings of the 7th Workshop on Middleware for Next Generation Internet Computing (MW4NG '12)*. ACM, New York, NY, USA, Article 1, 6 pages. DOI : <http://dx.doi.org/10.1145/2405178.2405179>